LA-UR-01-5895
Approved for public release;
distribution is unlimited.

Title: **Heat Pipe Transient Response Approximation**

Author(s): Robert S. Reid

Submitted to:

http://lib-www.lanl.gov/la-pubs/00796518.pdf

Heat Pipe Transient Response Approximation

Robert S. Reid

Los Alamos National Laboratory, Los Alamos, NM 87545
(505) 667-2626, FAX: -0600, rsr@lanl.gov

Abstract. A simple and concise routine that approximates the response of an alkali metal heat pipe to changes in evaporator heat transfer rate is described. This analytically based routine is compared with data from a cylindrical heat pipe with a crescent-annular wick that undergoes gradual (quasi-steady) transitions through the viscous and condenser boundary heat transfer limits. The sonic heat transfer limit can also be incorporated into this routine for heat pipes with more closely coupled condensers. The advantages and obvious limitations of this approach are discussed. For reference, a source code listing for the approximation appears at the end of this paper.

INTRODUCTION

Heat pipe transient response has been well studied, (Ambrose, 1991), (Bowman, 1994), (Cao, 1992), (Colwell, 1992), (Hall, 1994), (Issacci, 1991), (Jang, 1995), (Tournier, 1995), and (Tournier, 2001). The physical mechanisms are numerous and involved, especially if frozen startup is examined in any detail. Physics related to transient heat pipe operation can include: transition from free molecule to continuum flow in the vapor space, the migration of the melt front in capillary structures, mass transfer between the liquid and vapor regions, and compressibility effects. Entrainment of fluid from the wick, freezing of condensed vapor preventing fluid return to the evaporator, dewetting, and inadequate capillary pumping forces can limit heat pipe startup. Analytical techniques have been used to calculate frozen startup characteristics. Cao (1992) developed a heat pipe startup solution using analytical relations and a flat front assumption that is in some respects similar to the approach taken in this paper. Silverstein (1992) described a calculation approach that divides a heat pipe into evaporator, active, and inactive regions to find temperature history as the continuum front moves through the condenser.

This paper describes a response approximation for a fixed conductance alkali metal heat pipe to quasi-steady changes in evaporator heat transfer rate. A one-dimensional, lumped capacitance solution is coupled to analytical, laminar, incompressible, viscous limit and condenser boundary heat transfer relations. Although this approximation considers mechanisms essential to heat pipe transients, it ignores most important details and is ***not*** suited to rapid transients, to gas loaded heat pipes, or to heat pipes with strongly coupled condensers. Test data with analysis from first principle heat pipe codes best handle those cases. Figure 1 shows a nodal representation of a sodium-stainless steel heat pipe described in Reid (2001). The heat pipe tube is 2.54-cm OD and 2.22-cm ID. Final wick dimensions are 2.07-cm OD and 1.74-cm ID and form a crescent annular gap. Cartridge heaters radiate to fuel tubes that, in turn, radiate heat to the surroundings and conduct heat to the evaporator. The calculation divides the condenser into four axial nodes of equal length. Each node's mass center corresponds to a thermocouple measurement location. The heat capacity of the heaters, fuel tubes, heat pipe wall, and wick, radiation exchange with the environment, and vapor energy transport to the condenser are considered in the approximation.



FIGURE 1. Schematic Representation of Modeled Stainless Steel-Sodium Heat Pipe.



FIGURE 2. Condenser Heat Rejection Rate Versus Evaporator Temperature. (The dark dashed curve shows the sonic heat transfer rate limit, the solid curve the viscous heat transfer rate limit, and the light dashed curves indicate radiation limits for different active condenser lengths.)

FORMULATION

Figure 2 shows the condenser heat rejection rate versus evaporator temperature for the sodium heat pipe described in Reid (2001). Sodium vapor movement between the evaporator and condenser, internal energy held in the wall and wick, and radiant heat exchange between the heat pipe and the surroundings are the most significant mechanisms in the frozen startup power balance. The thermal energy transferred to condenser regions below 600 K is mostly held as internal energy in the wick, fluid, and wall. Energy exchange during sodium melting and solidification as well as axial and radial conduction through the wall are typically small and are ignored. Three conditions control the heat transfer rate to the condenser during alkali metal heat pipe frozen startup. Early on, the heat pipe is viscous limited: low saturation pressure constrains working fluid circulation. As vapor pressure increases with temperature, sonic flow or condenser coupling controls heat transfer rate to the condenser.

The unsteady one dimensional diffusion equation with a heating source term and radiation to the surroundings can be written:

$$C\frac{\partial T}{\partial t}=k\frac{\partial^2 T}{\partial x^2}+\dot{q}-G_R\left(T^4-T_\infty^4\right) \quad . \tag{1}$$

Writing Equation (1) in difference form about the cartridge heaters, fuel tubes, and evaporator node:

$$C_{CH}\,\frac{T_{CH}-T_{CH}^{p}}{\delta t}=\dot{q}_{ELECT}-G_{RI}\left(T_{CH}^4-T_{FT}^4\right), \tag{2}$$

$$C_{FT}\,\frac{T_{FT}-T_{FT}^{p}}{\delta t}=G_{RI}\left(T_{CH}^4-T_{FT}^4\right)-G_{RO}\left(T_{FT}^4-T_\infty^4\right)-G_C\left(T_{FT}-T_1\right), \text{ and} \tag{3}$$

$$C_1\,\frac{T_1-T_1^{p}}{\delta t}=G_C\left(T_{FT}-T_1\right)-G_{R,1}\left(T_1^4-T_\infty^4\right)-\dot{q}_{LIM} \quad . \tag{4}$$

The fuel tube to evaporator conductance, G_C, was assumed to be 12 W/K, based on measurement at a 900 K steady state. The minimum of the radiation, sonic, and viscous limits: $\dot{q}_{LIM}=\min\left(\dot{q}_R,\dot{q}_S,\dot{q}_V\right)$, determines the heat transfer rate $\dot{q}_2$ to the condenser. A power balance about the four condenser nodes is of the form:

$$C_i \frac{T_i - T_i^p}{\delta t} = \dot{q}_i - G_{R,i}\left(T_i^4 - T_\infty^4\right) \qquad i = 2...N_s \quad . \tag{5}$$

In the absence of external heat or work transfer into the node, thermodynamics requires $T_1 \geq T_i$. When $T_i > T_1$ is computed, T_i assumes the value T_1 and the power transferred to the next node is:

$$\dot{q}_{i+1} = \dot{q}_i - \frac{C_i}{\delta t}\left(T_i - T_i^p\right) - G_{R,i}\left(T_i^4 - T_\infty^4\right) \quad . \tag{6}$$

The Newton-Raphson method is used to solve Equations (2) through (5) for T_{CH}, T_{FT}, T_1, T_2, T_3, T_4, and T_5, respectively. The radiation coupling and the heat capacity of node 5 are adjusted for the presence of the cold liquid pool that accumulates at the condenser end of the heat pipe.

The minimum of the radiation, sonic, and viscous limits: $\dot{q}_{LIM} = \min\left(\dot{q}_R, \dot{q}_S, \dot{q}_V\right)$, is the heat transfer rate $\dot{q}_2$ to the condenser during startup. Boltzmann's equation establishes the heat emission rate from the condenser surface:

$$\dot{q}_R = \varepsilon\sigma\pi D_o\left(x_{POOL} - x_2\right)\left(T_1^4 - T_\infty^4\right). \tag{7}$$

The condenser starts at x_2 and continues to the edge of the liquid pool, x_{POOL}. The modeled heat pipe was not sonic limited during startup. For brevity, sonic limit calculations are omitted from this description.

At low temperature surface forces are undeveloped and evaporator vapor pressure circulates sodium through the vapor space and wick. A pressure balance at this condition is described by:

$$P_{sat} = \Delta P_{EVAP} + \Delta P_{COND} + \Delta P_{LIQ} \quad . \tag{8}$$

The analytical pressure drop relations used in this paper are listed in Woloshun (1988). Vapor pressure drop in the evaporator and condenser is divided into viscous and inertial terms. Radial mass injection stabilizes evaporator vapor flow. The laminar flow friction factor is $f = 16/\mathrm{Re}_D$, where Re_D is the axial Reynolds number, $\mathrm{Re}_D = 4\dot{q}_2 / \left(\pi d_v h_{fg} \mu_V\right)$. The inertial vapor plus viscous pressure drops for laminar incompressible flow in the evaporator region of a cylindrical heat pipe is:

$$\Delta P_{EVAP} = \left(\frac{\dot{q}_2}{h_{fg}}\right)^2 \frac{16 L_e f}{\pi^2 \rho_V d_V^5}\left[1 + \Psi\right] \quad , \tag{9}$$

the velocity profile correction factor, Ψ, a function of radial Reynolds number, $\mathrm{Re}_{r,e} = -\dot{q}_2 / \left(2\pi L_E h_{fg} \mu_V\right)$, is

$$\Psi = 0.61\,\mathrm{Re}_{r,e}\left(1 + 1/\left(3.6 + \mathrm{Re}_{r,e}\right)\right) \quad . \tag{10}$$

Condenser vapor flow is often turbulent from mass removal. Turbulent onset correlates with radial Reynolds number $\mathrm{Re}_{r,c} = -\dot{q}_2 / \left(2\pi L_C h_{fg} \mu_V\right)$. If $\mathrm{Re}_{r,c} < -2.25$, the viscous vapor pressure drop is calculated with $f = 16/\mathrm{Re}_D$ for laminar flow, $f = 0.079\,\mathrm{Re}_D^{-0.25}$ for $2{,}000 < \mathrm{Re}_D < 20{,}000$, and $f = 0.046\,\mathrm{Re}_D^{-0.20}$ for $\mathrm{Re}_D > 20{,}000$. For uniform heat removal and radial Reynolds number in the condenser, $\mathrm{Re}_{r,c} > -2.25$, the inertial pressure recovery for laminar incompressible flow in the condenser of a cylindrical heat pipe was given by Busse (1967):

$$\Delta P_C = P_{x=L_C} - \frac{64 \mu_V \dot{q}_2 L_C}{\rho_V h_{fg} \pi d_V^4}\,\mathrm{Re}_{r,c}\left(\frac{7}{9} - \frac{8}{27}\alpha + \frac{23}{405}\alpha^2\right)\left(1 - \frac{x - x_2}{L_C}\right)^2 \quad , \tag{11}$$

the velocity profile correction is $\alpha = 15/22\left(5 + 18/\mathrm{Re}_{r,c} + \left(\left(5 + 18/\mathrm{Re}_{r,c}\right)^2 - 44/5\right)^{0.5}\right)$. In the 1970s, Joe Kemme at the Los Alamos Scientific Laboratory, correlated cylindrical heat pipe condenser inertial pressure recovery for $\mathrm{Re}_{r,c} < -2.25$ to be a fraction of the inertial pressure at the condenser entrance (Woloshun, 1988):

$$\Delta P_C = \frac{\left(-\mathrm{Re}_{r,c} - 2\right)}{\rho_V\left(-1.23\,\mathrm{Re}_{r,c} + 2L_E/L_C\right)}\left(\frac{4\dot{q}_2}{h_{fg}\pi d_V^2}\right)^2 \quad . \tag{12}$$

The annulus provides the primary liquid sodium return path. Liquid pressure drop in the annulus is found using the Poiseuille flow equation based on hydraulic radius. An annotated Fortran-77 listing of the routine HPAPPX appears in this paper's appendices. Variable names and values of assumed constants can be inferred from this listing. SI units are used. Cartridge heater power versus time data for the comparison can be found in Reid (2001).



FIGURE 3. Comparison of Heat Pipe Surface Temperatures Measured (symbols) and Calculated (lines not carrying symbols) for a 6-Hour Test to 900 K. (Distance in legend is measured from the evaporator end cap.)

RESULTS AND CONCLUSION

Figure 3 shows the measured and calculated module temperature histories. Calculated fuel tube temperatures typically fell 20 K below measured values over much of the data set. Differing temperature definitions account for the discrepancy. Calculated fuel tube temperature is a surface average value whereas the thermocouple measurement was made near the peak fuel tube temperature. The measured and calculated evaporator temperatures agreed within a few degrees during startup. Condenser data lags measurement, but reach the knee of the startup temperature curve at about the right times. Data collection at discrete intervals accounts for much of the lag. The startup front moves steadily from the evaporator exit to the condenser end. The computed heat balance assumes an average nodal temperature and does not reflect the true active front progression. This situation might be remedied by increasing the number of axial nodes. The analytical pressure-drop relations in Equations (12) and (13) assume sodium circulation over the full condenser length during all startup phases. Actual circulation over that length occurs only after the pipe becomes completely isothermal. The actual startup pressure drop should be less than calculated by the analytical relations and the actual viscous heat transfer rate to the condenser should then be greater than that calculated.

Except for the 10-cm long condenser pool, the heat pipe became isothermal at 850 K, 7200 s into the startup. The calculated value at 1.09 m lagged the data some. Power was increased to the heat pipe until the evaporator reached 900 K. Calculated heat pipe surface temperatures agreed within 10 K of the data at this point. This is well within normal thermocouple measurement uncertainty. Power was then reduced to the evaporator at twice the application rate during startup. During shutdown the calculated and measured condenser cooling rates match. The calculation during shutdown misses the temperature gradient across the evaporator-fuel tube assembly and the condenser entrance. The omission of radial heat conduction from the model partly accounts for this discrepancy.

Overall, the approximation serves reasonably well, despite the application of steady-state pressure drop relations to data with a time changing component. Yet, such an approach may not be too restrictive: reactor thermal transients are often made in quasi-steady increments. Although no substitute for experiment or first principle transient heat pipe codes, this heat pipe startup approximation appears suited to those reactor core thermal hydraulic simulations that permit some sacrifice of accuracy to conciseness.

NOMENCLATURE

A	area (m^2)
C_i, C_{FT}, C_{CH}	heat capacity of heat pipe, fuel tubes, cartridge heater (J/K)
$\Delta P_C, \Delta P_E, \Delta P_L$	condenser, evaporator, and liquid pressure drops (Pa)
G_C	conductance between fuel tubes and evaporator (W/K)
$G_{R,i}$	radiation coupling of heat pipe surface nodes to surroundings (W/K^4)
G_{RI}, G_{RO}	fuel tube inside and outside radiation coupling (W/K^4)
N_S	number of heat pipe segments
$\dot{q}_i$	heat transfer rate between nodes (W)
$\dot{q}_{ELECT}$	electrical power to cartridge heaters (W)
$\dot{q}_c, \dot{q}_R, \dot{q}_S, \dot{q}_V$	conduction, radiation, sonic, and viscous heat transfer rates (W)
T_i, T_i^p	temperature at present time, preceding time (K)
T_{CH}, T_{FT}	temperature of cartridge heater, fuel tubes (K)
T_∞	radiation background temperature (K)
x_i	axial position from evaporator entrance (m)
$\varepsilon, \overline{R}, \sigma$	emittance (-), gas constant (J/kg-K), Stefan constant (W/m^2-K^4)

ACKNOWLEDGMENT

This work was supported by NASA Marshall Space Flight Center, Huntsville, AL.

REFERENCES

Ambrose, J. H., Chow, L. C., and Beam, J. E., "Detailed Model for Transient Liquid Flow in Heat Pipe Wicks," *Journal of Thermophysics and Heat Transfer,* 1991, **5** (4), pp. 532-538.

Bowman, W. J., Beran, P. S., "Implicit Heat Pipe Vapor Model," *Journal of Thermophysics and Heat Transfer*, 1994, **8** (1), 187.

Busse, C. A., "Pressure Drop in the Vapor Phase of Long Heat Pipes," in *Proceedings of the 6th Thermionic Conversion Specialist Conference*, published by the IEEE, 1967, pp. 391-398.

Cao, Y. and Faghri, A., "Closed-Form Analytical Solutions of High-Temperature Heat Pipe Startup and Frozen Startup Limitation," *Journal of Heat Transfer*, 1992, **114** (4), pp. 1028-1035.

Colwell, G. T. and Modlin, J. M., "Heat Pipe and Surface Mass Transfer Cooling of Hypersonic Vehicle Structures," *Journal of Thermophysics and Heat Transfer*, 1992, **6** (3), pp. 492-499.

Hall, M. L., Merrigan, M. A., and Reid, R. S., "Status Report on the THROHPUT Transient Heat Pipe Modeling Code," in *Proceedings of 12th Symposium on Space Nuclear Power and Propulsion*, edited by M. El-Genk, AIP Conference Proceedings 301, New York, 1994, pp. 965-970.

Issacci, F., Catton, I., and Ghoniem, N. M., "Vapor Dynamics of Heat Pipe Start-Up," *Journal of Heat Transfer*, 1991, **113** (4), pp. 985-994.

Jang, J. H., "Startup Characteristics of a Potassium Heat Pipe from the Frozen State," *Journal of Thermophysics and Heat Transfer*, 1995, **9** (1), pp. 117-122.

Reid, R. S., Sena, J. T., Martinez, A. L., 2001, "Sodium Heat Pipe Module Test for SAFE-30 Reactor Prototype," Space Technology and Applications International Forum-2001 (STAIF-2001), edited by M. El-Genk, AIP Conference Proceedings 552, pp. 869-874.

Silverstein, C. C, *Design and Technology of Heat Pipes for Cooling and Heat Exchange*, Taylor and Francis, Washington, D.C., 1992, pp. 251-253.

Tournier, J. M. and El-Genk, M. S., "HPTAM for Modeling Heat and Mass Transfers in a Heat Pipe Wick During Startup from a Frozen State," in *Proceedings of 12th Symposium on Space Nuclear Power and Propulsion*, edited by M. El-Genk, AIP Conference Proceedings 324, New York, 1995, pp. 123-134.

Tournier, J. M. and El-Genk, M. S., "Modeling of the Startup of a Horizontal Lithium Heat Pipe From a Frozen State," in *Proceeding of the 35th National Heat Transfer Conference*, NHTC01-11692, 2001, pp. 1-9.

Woloshun, K. A., Merrigan, M. A., and Best, E. D., *HTPIPE: A Steady-State Heat Pipe Analysis Program*, Los Alamos National Laboratory Manual, LA-11324-M, 1988.

APPENDIX 1: FORTRAN-77 SOURCE LISTING OF PROGRAM HPAPPX.F

```
                main program          PROGRAM HPAPPX
              dimension arrays        REAL C(5),GR(5),Q(6),T(5),TP(5),X(6)
            SI unit constants         DATA CCH,CP,DO,DI,E,GC,NS,PI/1514.,560.,.0254,.022,0.4,12.,5,3.14/
            SI unit constants         DATA RHOS,SIGMA,TIMEP,TI,TCHP,TFT/7900.,5.67E-8,0.,296.,296.,296./
            SI unit constants         DATA ECH,EFT,DCH,DIFT,TCH,TFTP,XP/.4,.4,.0189,.022,296.,296.,1.1/
       initial power to nodes         DATA (Q(I),I=1,6)/0.,0.,0.,0.,0.,0./
        previous temperatures         DATA (TP(I),I=1,5)/295.,296.,295.,296.,295./
         initial temperatures         DATA (T(I),I=1,5)/295.,296.,295.,296.,295./
              node boundaries         DATA (X(I),I=1,6)/.0000,.4300,.6095,.8125,1.002,1.200/
               input data file        OPEN(UNIT=1,FILE='INPUT',STATUS='OLD')
              output data file        OPEN(UNIT=2,FILE='OUTPUT',STATUS='NEW')
          wall sectional area         AC=PI/4.*(DO**2-DI**2)
          heater surface area         ACH=4.*PI*(DCH*.0254)*(X(2)-X(1))
        fuel tube inside area         AFTI=4.*PI*DIFT*(X(2)-X(1))
       fuel tube outside area         AFTO=(8.*.0254)*(X(2)-X(1))
         liquid pool fraction         POOL=(XP-X(5))/(X(6)-X(5))
           fuel heat capacity         CFT=4.*CP*RHOS*PI/4.*(DO**2-DIFT**2)*(X(2)-X(1))
        NS=number of segments         DO 10 I=1,NS
           node heat capacity      10 C(I)=CP*RHOS*AC*(X(I+1)-X(I))
          inactive pool region        C(5)=C(5)*POOL
   E=total hemisp. emittance          DO 20 I=1,NS
    node radiation couplings      20 GR(I)=E*SIGMA*PI*DO*(X(I+1)-X(I))
         evaporator coupling          GR(1)=GR(1)/PI
       condenser end coupling         GR(5)=GR(5)*POOL
         heater-fuel coupling         GRI=SIGMA*ACH/(1./ECH+ACH/AFTI*(1./EFT-1.))
        fuel tubes to ambient         GRO=SIGMA*EFT*AFTO
    read number of time steps         READ(1,*)NT
              main time loop          DO 30 N=1,NT
   time and heater power read         READ(1,*)TIME,QELECT
              time increment          DT=TIME-TIMEP
     CCH=heater heat capacity         BB=CCH/GRI/DT
                                      CC=TFTP**4+BB*TCHP+QELECT/GRI
                                      DO 40 J=1,5
           heater temperature      40 TCH=TCH-(TCH**4+BB*TCH-CC)/(4.*TCH**3+BB)
  CFT=fuel tube heat capacity         BB=(GC+CFT/DT)/(GRI+GRO)
     GC=fuel tube-evap. cond.         CC=(GRI*TCH**4+GRO*TI**4+GC*T(1)+CFT/DT*TFTP)/(GRI+GRO)
                                      DO 50 J=1,5
            fuel temperature      50 TFT=TFT-(TFT**4+BB*TFT-CC)/(4.*TFT**3+BB)
   fuel-evaporator conduction         Q(1)=GC*(TFT-T(1))
                                      BB=C(1)/GR(1)/DT
       TI=ambient temperature         CC=TI**4+BB*T(1)+(Q(1)-Q(2))/GR(1)
                                      DO 60 J=1,5
       evaporator temperature      60 T(1)=T(1)-(T(1)**4+BB*T(1)-CC)/(4.*T(1)**3+BB)
              radiation limit         QR=E*SIGMA*PI*DO*(XP-X(2))*(T(1)**4-TI**4)
                viscous limit         CALL QVISC(T(1),QV)
   power to condenser heating         Q(2)=AMIN1(QR,QV)
   power to condenser cooling         IF(TP(1).GT.T(1))Q(2)=QR+CP*RHOS*AC*(XP-X(2))*(T(2)-TP(2))/DT
               condenser loop         DO 70 I=2,NS
                                      BB=C(I)/GR(I)/DT
                                      CC=TI**4+BB*T(I)+Q(I)/GR(I)
                                      DO 80 J=1,5
       condenser temperatures      80 T(I)=T(I)-(T(I)**4+BB*T(I)-CC)/(4.*T(I)**3+BB)
            node fully active         IF(T(I).GT.T(1))THEN
         Second Law constraint        T(I)=T(1)
            power to next node        Q(I+1)=Q(I)-C(I)/DT*(T(I)-TP(I))-GR(I)*(T(I)**4-TI**4)
                                      ELSE
         vapor not at node yet        Q(I+1)=0.
                                      ENDIF
        end of condenser loop     70 CONTINUE
                                      DO 90 J=1,NS
  save heat pipe temperatures     90 TP(J)=T(J)
          store previous time         TIMEP=TIME
  store fuel tube temperature         TFTP=TFT
     store heater temperature         TCHP=TCH
          send results to file        WRITE(2,1)TIME,TCH,TFT,(T(J),J=1,5),Q(1),Q(2)
                                     1 FORMAT(10(1X,F6.0))
             end of time loop     30 CONTINUE
                                      END
```

APPENDIX 2: FORTRAN-77 SOURCE LISTING OF HPAPPX.F SUBROUTINES

```
     viscous limit subroutine          SUBROUTINE QVISC(T,QV)
        initial viscous limit          QV=0.
           viscous limit loop       10 CONTINUE
  1-W increment viscous limit          QV=QV+1.
    pressure drops calculated          CALL DELTP(QV,T,DPE,DPC,DPL)
total heat pipe pressure drop          DPTOT=DPE+DPC+DPL
       SI pressure polynomial          P=((((.33132E-13*T-.18721E-9)*T+.42507E-6)*T-.49438E-3)*T+
                                      1 .30911)*T-.75842E2
         SI saturation pressure        PSATO=EXP(P)
      not at viscous limit yet         IF(DPTOT.LT.PSATO)GO TO 10
                                       RETURN
                                       END

     pressure drop subroutine          SUBROUTINE DELTP(Q,T,DPE,DPC,DPL)
                SI constants           DATA PI,RBAR,RV,XLC,XLE,XMW/3.14,361.,.0087,.77,.43,23./
             effective length          XLEFFL=(XLE+XLC)/2.
    SI latent heat polynomial          HFG=((((-.12942E-8*T+.62049E-5)*T-.11117E-1)*T+.9097E1)*T-
                                      1 .41858E4)*T+.53352E7
                mass flow rate         QHFG=Q/HFG
  SI liquid dynamic viscosity          XMUL=((((-.20311E-17*T+.10958E-13)*T-.23417E-10)*T+.24986E-7)*T
                                      1 -.13598E-4)*T+.32842E-2
             annulus thickness         A=(.0220-.0207)/2.
             SI liquid density         RHOL=((((.38804E-13*T-.19426E-9)*T+.37279E-6)*T-.34209E-3)*T
                                      1 -.88062E-1)*T+.99142E3
Poiseuille flow pressure drop          DPL=6.0*XMUL*QHFG*XLEFFL/(PI*RV*A**3*RHOL)
              vapor space area         AV=PI*RV**2
          vapor space diameter         DV=SQRT(4.*AV/PI)
SI vapor viscosity polynomial          XMUV=((((.76982E-20*T-.3217E-16)*T+.49847E-13)*T-.36443E-10)*T
                                      1 +.2506E-7)*T+.68708E-5
         axial Reynolds number         REYV=4.*QHFG/PI/XMUV/DV
    evaporator Reynolds number         RRN=QHFG/2./PI/XLE/XMUV
          velocity correction          PSI=0.61*RRN+0.61*RRN/(3.6+RRN)
  SI vapor density polynomial          RHOV=EXP(((((.34434E-13*T-.1925E-9)*T+.43234E-6)*T-.49716E-3)*T+
                                      1 .30672)*T-.86671E2)
                                       AVIS=16./REYV*XLE/DV
                                       BETA=AVIS*PSI
         viscous evaporator dp         DPVE=AVIS*(QHFG/AV)**2/RHOV
        inertial evaporator dp         DPIE=(QHFG/AV)**2*BETA/RHOV
           total evaporator dp         DPE=DPIE+DPVE
condenser rad Reynolds number          RREYC=-QHFG/2.0/PI/XLC/XMUV
                                       RM2T=QHFG/AV/RHOV/SQRT(RBAR*T/XMW)
                                       VCI=RM2T*(RBAR*T/XMW)**.5
       laminar friction factor         F=16.0/REYV
     turbulent friction factor         IF(REYV.GT.2000.)F=0.079/REYV**.25
     turbulent friction factor         IF(REYV.GT.20000.)F=0.046/REYV**.2
          viscous condenser dp         DPVC=4.0*F*(XLC/2)*RHOV*VCI**2/(4.0*RV)
  Busse condenser dp relation          IF(RREYC.GT.-2.25)THEN
                                       B=15.0/22.0*(5.0+18.0/RREYC+SQRT((5.0+18.0/RREYC)**2-44.0/5.0))
                                       COEFFI=-RREYC*(7.0/9.0-8.0*B/27.0 +23.0*B**2/405.0)
          Busse pressure drop          DPIC=COEFFI*4.0*XMUV*VCI*XLC/RV/RV
  Kemme condenser dp relation          ELSE
   no adiabatic section here!          LPARAM=(2.*XLE)/XLC
    inertial pressure recovery         RECOV=(RREYC+2.)/(1.23*RREYC-LPARAM)
 Kemme inertial pressure drop          DPIC=-RECOV*RHOV*VCI**2
                                       ENDIF
total condenser pressure drop          DPC=DPIC+DPVC
                                       RETURN
                                       END
```